
May this presentation improve the security of organisations great and small.

Detection Technique Deep Dive

• This presentation builds upon .conf 2018 SEC1038 Detection Technique Deep Drive 
(https://www.splunk.com/en_us/resources/videos/detection-technique-deep-dive.html)
which I recommend watching after this presentation for completeness.

• The case studies you see in this session can be found in the app: 
https://splunkbase.splunk.com/app/4209

• So we can test our SPL™, every time you run a search using the spike `casestudy` macros, 
they generate new data that has roughly the same statistical properties.

• We use the Set Operations Technology Add-On (https://splunkbase.splunk.com/app/3516/)

https://www.splunk.com/en_us/resources/videos/detection-technique-deep-dive.html
https://splunkbase.splunk.com/app/4209
https://splunkbase.splunk.com/app/3516/

SPL™ for Detection
The only commands you’ll ever need

* Provided by the Set Operations Technology Add-on

search

lookup

stats

eval

 eventstats

 where

 bin

mvexpand

 tstats

 streamstats

 setop

 distinct*

Spikes
Median Absolute Deviation

    | ...
    | eventstats median(y) as median
    | eval absolute_deviation=abs(y-median)
    | eventstats median(absolute_deviation) as median_absolute_deviation
    | where y>median+median_absolute_deviation*#

• Let your data guide you to the appropriate # number

• Use fields such as src_ip or user as ‘by’ fields in the eventstats as required

Spikes
Data with Seasonality

Detecting spikes in data with seasonality is out of this presentation scope, however:

• Consider fields such as date_wday and date_hour as ‘by’ fields to account for seasonality 
when using statistical methods such as Standard Deviation or Median Absolute Deviation.

 
• For more complex seasonality, try `brown_manoeuvre(y)` from the .conf 2018 session.

First-Time Events
When a new value is seen in a field or combination of fields

    | ...
    | eventstats count by fieldA ... fieldn
    | where count == 1

In practice, we usually use streamstats rather than eventstats and add 
_time>relative_time(now(),"-1h") to the where predicate. Needless to say, use fields such 
as src_ip and user as ‘by’ fields.

In some cases, a higher threshold (such as < 3) may be necessary to prevent false-negatives. 
Consider finding rare rather than unique values using something like:

    | ...
    | streamstats count as datapoint
    | streamstats count by fieldA
    | where count/datapoint < 0.01 AND _time>relative_time(now(),"-1h")

First-Time Events
Distinct fields

• If you want to determine which fields contain unique values, the distinctfields and 
distinctstream commands are used.

• The cardinality of the distinctfields field can then be used to measure behaviour change.

• We have limited time so will focus more on sequences in this session. For more on the use 
of distinct fields, please see the .conf 2018 presentation and SetOps documentation: 
https://github.com/doksu/setops/wiki#usage

https://github.com/doksu/setops/wiki#usage-1

Simple Sequences
Detecting if “this then that” happens



© 2022 SPLUNK INC.

Splunk .conf22 Template  |  TMPLT-FY23-101  |  v1

Simple Sequences
Detecting if “this then that” happens

    | ...
    | streamstats current=f last(fieldA) as last_fieldA by ...
    | where fieldA != last_fieldA AND _time>relative_time(now(),"-1h")

T1071.004 DNS Command and Control
Possible C2 Beacon Detected Through Domain Parking

 | tstats `summariesonly` values("DNS.answer") as answer FROM
   datamodel=Network_Resolution WHERE (nodename=DNS AND DNS.src="10.*" AND
   DNS.query!="*.in-addr.arpa") BY _time "DNS.src", "DNS.query" span=5m

 | `drop_dm_object_name("DNS")`
 | mvexpand answer
 | where match(answer,"^\d+\.\d+\.\d+\.\d+$")
 | streamstats current=f window=1 global=f earliest(_time) as previous_time,
   last(answer) as previous_answer by query

 | where answer!=previous_answer
   AND match(answer,"^127\.") XOR match(previous_answer,"^127\.")
 | eval minutes_difference=round((_time-previous_time)/60)
 | lookup dnslookup clientip AS src OUTPUT clienthost AS src_host

Simple Sequences
Using Set Operations

    | ...
    | streamstats current=f values(fieldA) as previous_values_fieldA by ...
    | setop op=difference fieldA previous_values_fieldA
    | where mvcount(difference) > 1 AND _time>relative_time(now(),"-1h")

T1021.001 Lateral Movement with RDP
Multiple Servers RDPed to for First Time by User

 source="WinEventLog:Security" EventCode=4624 Logon_Type=10

 | bin _time span=1d
 | eval user=lower(user), ComputerName=lower(ComputerName)
 | stats values(ComputerName) as dest_host, values(src_ip) as src_ip
   by _time, user
 
 | streamstats current=f values(dest_host) as previous_dest_host by user
 | setop op=difference dest_host previous_dest_host
 | eval difference_count=mvcount(difference)

 | where difference_count>1 AND _time>relative_time(now(),"-24h")
 | eval risk_object=user, risk_object_type="user",
   risk_score=difference_count*20

Complex Sequences
An example of what can be achieved in SPL™

T1110 Brute Force Attack
Successful Authentication by User After High Number of Failures

 | tstats `summariesonly` count FROM datamodel=Authentication WHERE
   nodename=Authentication AND Authentication.action IN
   ("success","failure") NOT Authentication.user IN ("unknown", "root")
   BY _time, "Authentication.user", "Authentication.action" span=1s
 
 | `drop_dm_object_name("Authentication")`
 | streamstats current=f last(action) as last_action by user
 | streamstats count(eval(action=="success" AND last_action=="failure"))
   as auth_series by user
 | streamstats sum(eval(if(action=="failure", count, 0))) as sum
   by user, auth_series
 | streamstats current=f last(sum) as last_sum by user
 
 | where action=="success" AND last_action=="failure" AND last_sum>50
 | eval risk_object=user, risk_object_type="user",
   risk_score=round(last_sum/100)*25

Highly Complex Sequences

• The need to detect highly complex sequences is as rare as it is complex.

• State machines can be used to detect if a complex sequence of events has occurred or if a 
certain depth in a state machine has been reached.

• State machines are outside the scope of this presentation but if you’re interested, please 
see: https://www.youtube.com/watch?v=5ToTZYm5bjw

https://www.youtube.com/watch?v=5ToTZYm5bjw

Thank You


